ГЛАВА 1. ОСНОВЫ ПРОГРАММИРОВАНИЯ
§ 1. Строковый тип данных
1.1. Строковые величины

До сих пор мы рассматривали программы, предназначенные для обработки только числовых данных. Многие задачи требуют умений работы с текстовыми данными, которые в языке Паскаль представлены в виде строк.

Величина строкового типа представляет собой последовательность знаков расширенной таблицы кодов ASCII (строку текста). Идентификатором типа является зарезервированное слово STRING. Объем памяти, занимаемый значением строкового типа, может составлять от 1 до 256 байт в зависимости от количества знаков в его составе.

Таблица кодов ASCII – американский стандарт кодов для обмена информацией. Таблица содержит строчные и прописные буквы латинского алфавита, цифры, разделители и другие знаки. В Приложении 1 приводится базовая и расширенная таблица кодов ASCII, содержащая также кириллицу (буквы русского алфавита).
Строковая константа  чаще всего имеет вид последовательности произвольных знаков таблицы ASCII, ограниченной апострофами. Примеры строковых констант: '123', '' (пустая строка), 'hello', 'A'. В словах, содержащих апостроф, этот знак записывается дважды, например, 'Мар''iна Горка'. 

Именованные константы описываются в разделе констант программы, например, 

Const b='Класс 11я';  

{строковая константа b}

Знаки строки считаются пронумерованными по порядку слева направо, начиная с 1. Их общее количество называется длиной строки. Так, длина описанной выше константы b равна 9; порядковый номер знака 'К' равен 1, 'л' – 2, …, 'я' – 9; знаку ' ' (пробел) соответствует номер 6. 
Длина пустой строки '' (она состоит из двух рядом стоящих апострофов) считается равной нулю. 

Строковая константа должна быть записана без переносов. При этом ее длина на экране монитора может занимать не более 127 знаков (включая ограничивающие апострофы).

Переменные строкового типа описываются в разделе переменных программы, например,  Var Text, L: String;   Здесь описаны переменные Text и L строкового типа с ограничением длины 255 знаков.

 * Для более коротких строк максимальную длину программист может задать самостоятельно. Например, согласно описанию 

Var A: String[7]; B:String; максимальные длины строк A и B составляют соответственно 7 и 255 знаков. 
1.2. Операции со строковыми величинами

Над строковыми величинами можно выполнять следующие операции: сравнение, объединение, ввод, вывод, присваивание, запись в составе выражений.

Сравнение строк реализуется через последовательное сравнение их соответствующих знаков: если очередной знак строки Х имеет больший код в таблице символов ASCII, чем соответствующий знак строки Y, считают, что X>Y. Например, '0'<'1'<'2', 'Fa'<'Fabel'<'Fach'<'Fest'< 'Feuer'<'Feuer '<'Maus'<'so'. Результат операции сравнения – величина логического типа.

	Операция
	Результат
	Операция
	Результат
	Операция
	Результат

	'123'<'13'
	True
	'12'>'2'
	False
	'mam'>'mab'
	True

	'abba'>='ab'
	True
	'car'='car'
	True
	'cat'<='c'
	False


Операция объединения (склеивания) обозначается символом «+» и позволяет присоединить одну строку к другой. Длина полученной строки равна сумме длин строк-операндов и не может быть более 255 знаков.

Пример 1. Найти результат операции объединения строк A и B, если  А='123', B='4'. 

Результат операции объединения строк зависит от порядка их записи в операторе присваивания: A+B='1234', B+A='4123'. 

 Использование операции «+» позволяет избежать ограничения выводимой на экран монитора длины строки (до 127 знаков). Для получения строки длиной до 255 знаков нужно записать нескольких более коротких строк, объединяя их с помощью операции «+». 

Строковые величины могут использоваться в составе оператора присваивания. Так, пусть переменная Y описана как Var Y: String[5]; В результате выполнения оператора присваивания Y:='HELLO,BABY!'; переменная Y получит значение 'HELLO' (лишняя часть значения теряется). 

Пример 2. Составить программу, позволяющую определить, решил ли школьник контрольную работу.

Пусть OC – оценка ученика по контрольной работе. Будем считать, что контрольная работа решена, если 5≤ОС≤10.

Program KONTRAB;

  Const X=’контрольная’;

        Y=’ не’; Z=’ решена’;

  Var OC: Byte; RES: String;

Begin
  Write(’Оценка по контрольной работе=’);

  Readln(OC);

  If (OC>=5)and(OC<=10) Then RES:=X+Z

       Else RES:=X+Y+Z;

  Writeln(RES);

End.

При OC=7, например, значение строки RES будет равно 'контрольная решена'; при ОС=3 RES='контрольная не решена'.

Так как знаки строки считаются пронумерованными, можно обратиться к любому знаку этой строки, указав в квадратных скобках после имени переменной порядковый номер (индекс) знака. Использование такой индексированной переменной позволяет читать и изменять отдельные знаки строки. Например, для Х='informatics': X[1]='i',X[5]='r',X[7]='a'. 

 1. Пусть X='forma'. После выполнения оператора присваивания X[2]:='i'; строковая переменная X изменит свое значение: X='firma'.

* 2. В процессе выполнения программы при использовании индексированной переменной значение индекса сравнивается с длиной строки. Знаки, индекс которых больше длины строки, считаются доступными для индексирования, однако являются неопределенными; операции над ними недопустимы. Например, пусть строка S описана как Var S:String[10];. В результате присваивания: S:='12+34'; фактическая длина строки окажется  равной 5. Обращение S[7] некорректно (результат не определен), обращение S[15] приведет к ошибке компиляции. *

Пример 3. Написать программу простого диалога с пользователем.

Программа должна предложить ввести имя и вывести приветствие, обращаясь к пользователю по имени.

На языке Паскаль программа может выглядеть так:

Program Privet_1;

  Var T: String;

Begin
  Write('Как тебя зовут? ');

  Readln(T);

  Writeln('Здравствуй,',T,'!');

End.
* Пример 4. Написать программу выборочного диалога.

Программа должна предложить ввести имя и выбрать пол (мужской, женский) и в зависимости от выбранного пола добавить к приветствию обращение «дорогой» или «дорогая». При неверном выборе признака пола программа должна вывести сообщение «Не понял юмора». 

Для ввода признака пола будем использовать строковую переменную P. Получим следующую программу: 
Program Privet_2;

   Var T, P: String;

Begin
   Write('Как тебя зовут? ');

   Readln(T);

   Writeln('Кто ты: Юноша (m)');

   Writeln('Девушка (w)');

   Readln(P);

   If (P='m') or (P='M') Then 

     Writeln('Здравствуй, дорогой ',T,'!')

     Else If (P='w') or (P='W') Then 

         Writeln('Здравствуй, дорогая ',T,'!')

        Else Writeln('Не понял юмора!');

End.

1.3. Стандартные подпрограммы

Познакомимся со стандартными подпрограммами, предназначенными для работы со строковыми величинами. Покажем их работу на примере строки T='Good bye':

	Подпрограмма
	Назначение
	Пример

	Length(X)
	Функция для определения длины строки Х (количества знаков)
	D:=Length (T); ( D=8

	Copy(X,K,N)
	Функция для копирования подстроки Х длиной N знаков, начиная со знака с номером K
	B:=Copy(X,7,2);(B='yе'

C:=Copy(X,4,1);( C='d'

D:=Copy(X,9,1);( D=''

E:=Copy(X,6,9);(E='bye'

	Pos(X,Y) 


	Функция для поиска первого вхождения подстроки Х в строку Y (результат – номер знака в Y, с которого начинается Х, или нуль, если подстрока не обнаружена)
	K:=Pos(S,T);

При S='o':  K=2

При S='a': K=0

При S='odb': K=0

При S='od b': K=3

	Delete(X,K,D);
	Процедура удаления из строки Х подстроки длиной D знаков, начиная с позиции K
	Delete(T,6,3); 

Insert('luck',T,6); 

вначале из T удаляется подстрока 'bye', а затем вставляется подстрока 'luck' начиная со знака с номером 6 ( T='Good luck'

	Insert(X,Y,K); 


	Процедура вставки подстроки Х в строку Y, начиная со знака с номером К
	


Использование операции объединения и функции копирования подстроки позволяет получать из заданных строк новые строки. Например, пусть А='программирование', В='идентификатор', C='реформа', тогда строка 'информатика и программирование' может быть получена как результат выполнения операций: 
B[1]+B[4]+Copy(C,3,5)+B[5]+Copy(B,8,3)+ ' '+B[1]+ ' '+A.

Пример 1. Написать программу определения количества букв в имени. 
Программа предлагает ввести имя пользователя T и выводит значение K – количество букв в имени, первую T[1] и последнюю T[K] буквы имени.

Программа будет такой:

Program Privet_3;

   Var T: String; K: Byte;

Begin
   Write('Как тебя зовут? ');

   Readln(T);  

   Writeln ('Здравствуй, ',T,'!');

   K:=Length(T);

   Writeln('Число букв в имени=',K);

   Writeln('1-я буква имени=',T[1]);

   Writeln('последняя буква=',T[K]);

End.
Пример 2. Написать программу определения количества букв «а» в заданном предложении. Известно, что предложение заканчивается точкой. 

Пусть T – заданное предложение. Введем переменную A для хранения количества букв «а» в предложении T. Будем посимвольно просматривать текст T в направлении слева направо, сравнивая каждый знак введенной строки с буквой «а». Значение A следует увеличить на 1, если выполнится условие T[i]='a' или T[i]='A'.  

Программа будет такой:    
Program KOL_A;

   Var T: String; 

       A,i: Byte;

Begin
  Writeln('Введите предложение');

  Readln(T); 

  A:=0;
  For i:=1 To Length(T) do {i – номер очередного знака}

  Begin 

      If (T[i]='a') or (T[i]='A') Then A:=A+1;

  End;

  Writeln('Количество букв “а”=',A);

End.

* Пример 3. Написать программу «Поле чудес».
Программа должна предложить угадать задуманное слово. Пользователь вводит некоторую букву, программа должна «открыть» ее в соответствующих позициях слова (если она есть в задуманном слове).

Пусть задуманное слово S задается в разделе констант. В этом случае обращение к определенному знаку строки осуществляется через стандартную функцию Copy. 

Решение задачи распадается на следующие этапы:

1) вывод на экран шаблона слова в виде строки знаков «-», причем количество этих знаков должно совпадать с длиной слова;

2) ввод пользователем буквы B;

3) вывод этой буквы на экран в соответствующую позицию, если такая буква есть в слове (для проверки наличия буквы B в слове S будем просматривать слово, начиная с первой буквы к последней);
4) выполнение пунктов 2)-3) нужно продолжить до тех пор, пока не будет отгадано все слово. 

Программа будет такой:

Program POLE;
   Uses Crt;

   Const S='mathematics';
   Var B: String; i,K,D: Byte;

Begin
   Writeln('Угадай слово');
   D:=Length(S);

   For i:=1 To D do  

{Вывод шаблона слова}

   Begin

     GotoXY(i*2,10); Write ('-');

   End;

   Writeln;

   K:=0;         {K-количество угаданных букв}

   While K<>D do  {Пока не угаданы все буквы,}

   Begin

      GotoXY(10,15); 
      Write('Введите букву ');  

      GotoXY(25,15); 
      Write(' ');  {стираем предыдущую букву,}

      GotoXY(25,15);
      Readln(B);       {вводим новую букву,}

      For i:= 1 to D do 

                {проверяем, есть ли буква в слове}

       If Copy(S,i,1)=B Then {есть буква}
       Begin            

         K:=K+1; 

         GotoXY(i*2,10); Write(B);

       End;

   End;

   GotoXY(10,20); 

   Writeln('Поздравляю с победой!');

End.

 Эта программа будет работать неправильно, если пользователь повторно введет уже отгаданную букву. Подумайте, как исправить эту погрешность.
*Пример 4. Написать программу замены в заданном тексте всех сочетаний «ой» на сочетание «ая».

Метод 1. Будем производить требуемую замену непосредственно в заданной строке S: после обнаружения фрагмента «ой» удалим его и вставим «ая». 

Так как требуется просмотр всей строки, в программе можно использовать цикл с параметром: вначале будем копировать 1-й-2-й знаки, затем 2-й-3-й, 3-й-4-й и т.д., и, наконец, предпоследний и последний знаки строки S. Переменную k будем использовать для задания номера левого знака из копируемой подстроки.

Program SAM_1;

   Var S: String; k: Byte;

Begin
   Writeln('Введите строку'); Readln(S);  

   For k:=1 To Length(S)-1 do

     If Copy(S,k,2)='ой' Then 

     Begin

       Delete(S,k,2); Insert('ая',S,k);

     End;

   Writeln('текст после замены: ',S);

End.

Метод 2. В отличие от метода 1 для нахождения очередной подстроки «ой» в строке S будем использовать стандартную функцию POS. Продвижение по строке в данном случае не требуется: функция будет использоваться столько раз, сколько в S имеется подстрок «ой». Так как количество «ой» заранее не известно, уместно использование цикла While (возможно, что в S не окажется ни одной подстроки «ой»).
Программа будет такой:

Program SAM_2;

   Var S: String; k: Byte;

Begin

   Writeln('Введите строку'); Readln(S);  

   k:=Pos('ой',S);

   While k<>0 do

   Begin

      Delete(S,k,2); Insert('ая',S,k);

      k:=Pos('ой',S);

   End;

   Writeln('текст после замены: ',S)

End.
* Пример 5. Написать программу вывода текста с эффектом «падающие буквы».  
После ввода строки T выведем ее в первую строку экрана. Будем постепенно «опускать» первый знак строки T с первой строки экрана в последнюю, стирая его при этом в предыдущей строке. Аналогично поступим со 2-м, 3-м и остальными знаками строки T. 

Пусть В – очередной знак строки Т, S – номер строки экрана, в которой отображается B, n – номер знака строки T.

Программа будет такой:

Program Padenie;

   Uses Crt;   

   Var T,B : String;  

       n,S : Byte;

Begin
   Writeln('Введите строку'); 

   Readln(T); 




{Ввод строки}

   ClrScr;





{Очистка экрана}
   GotoXY(1,1); 
   Write(T);
{Вывод строки в левый верхний угол экрана}

   For n:=1 To Length(T) do  {n-номер очередного знака}

   Begin
     B:=T[n];


  {B – очередной знак}

     For S:=2 to 25 do     {на экране 25 строк}

     Begin
       GotoXY(n,S-1); {Вывод пробела – стирание знака}

       Write(' ');
 {в предыдущей строке экрана.}

       GotoXY(n,S);   {Вывод очередного значения B} 
       Write(B);      {в текущей строке экрана.}

       Delay(1000);   {Задержка изображения}

     End;

   End;

End.
Пример 6. «Шифрование». Написать программу шифрования текста сообщения, заменяя каждый знак следующим за ним, а последний – первым (например, если введен текст 'SECRET', то результат работы программы –  'ECRETS').
Пусть Text – заданный текст, New_Text – текст после шифрования. Вначале присвоим New_Text первый знак строки Text, и будем присоединять слева по одному знаку из заданной строки при движении справа налево.
Program SHIFR;

  Var Text,New_Text: String;

      D,i: Byte;

Begin
  Writeln('Введите текст для шифрования');

  Readln(Text);

  D:=Length(Text);

  New_Text:=Text[1];

  For i:=D downto 2 do

    New_Text:=Text[i]+New_Text;

  Writeln('Зашифрованный текст:');
  Writeln(New_Text);

End.

1. Как объявить  величину строкового типа? 

2. Какова максимально возможная длина строки? 

3. Какие операции можно выполнять над строковыми величинами?

4. Какие процедуры и функции определены для величин строкового типа?

 Упражнения

1. Напишите программу, которая заменяет в тексте Т знак А на знак В (значения Т,  А и В должны вводиться с клавиатуры).

2. Напишите программу, которая подсчитывает количество слов в телеграмме. Известно, что слова разделяются только одним пробелом; текст телеграммы заканчивается точкой.
*3. Напишите программу шифрования текста заданного сообщения, вставляющую после каждой буквы слог «ва» (например, для текста “прикол” получите ответ “пварваивакваовалва”). 

* 1.4. Преобразование строковых данных 

Для преобразования строковых величин в числовые и обратно используются стандартные процедуры:

Val (S, X, C); 

Преобразование строки S в значение числовой переменной Х; целочисленная переменная C при этом играет роль признака возможности такого преобразования: если C=0, преобразование успешно; в противном случае значение C равно позиции ошибочного знака в строке.

Str (X, S);


Преобразование целого (или вещественного) числа Х в строку S.

Благодаря этим процедурам строковые величины можно преобразовать в числовые, к которым применимы арифметические операции, а числовые данные – в строковые и применять к ним строковые функции. 

Выполнение процедуры Str(X,S); рассмотрим на примере:

	Описание
	X
	S
	Длина строки

	Var X:Integer;S:String; 
	12345 
	'12345'
	5

	
	-12345 
	'-12345'
	6

	Var X:Real;S:String;
	12345 
	' 1.2345000000E+04'
	17

	
	-12345 
	'-1.2345000000E+04'
	17

	
	34.5
	' 3.4500000000E+01'
	17


Выполнение процедуры   Val(S,X,Code); рассмотрим на примере:

	Описание
	S
	X
	C

	Var S:String;X,C:Integer;
	'170'
	170
	0

	
	'5.5'
	0
	2

	Var S:String;X:Real;C:Integer;
	'17E3'
	1.7000000000E+04
	0

	
	'-0.25'
	-2.5000000000E-01
	0

	
	'93A5'
	0.0000000000Е+00
	3


Пример 1. Написать программу определения суммы цифр заданного натурального числа.   

Рассматривая число как строку и преобразуя каждый знак в число с помощью процедуры Val, можно найти сумму цифр достаточно «длинного» числа. 

Пусть A – заданная строка, S – искомая сумма, k – номер знака (цифры), С – результат преобразования знака в число.

Полагаем, что данные корректны, и после преобразования сразу выполняем суммирование цифр.

Программа будет такой:

Program SUMM;

   Var A: String; 

       S,C,k,cod: Integer;

Begin

    Write('Число='); Readln(A);  

    S:=0;

    For k:=1 to Length(A) do

    Begin

      Val(A[k],C,cod);

      S:=S+C;

    End;

    Writeln('S=',S);

End.

Пример 2. Вводится строка в виде: А*B=. Здесь A, B – натуральные числа; * – символ операции (+ или -). Требуется написать программу, которая позволит вычислить и вывести на экран результат выполнения операции.

Пусть S – заданная строка, представленная в виде арифметического выражения; R – искомое значение арифметического выражения.
Этапы решения задачи:

1) ввод строки S;
2) определение Р – номера знака в строке S, где находится знак операции (согласно условию, '+' или '-');

3) копирование из строки S подстрок N и K, где N – подстрока левее знака операции, K – подстрока правее знака операции; например, если S='15+435', то N='15', K='435';

4) преобразование строк N и K соответственно в значения числовых переменных А и В;

5) распознавание арифметической операции и ее выполнение.

Будем считать, что данные корректны: строка S содержит знак арифметической операции ('+' или '-').
Программа решения задачи имеет вид:

Program ARIFM;

  Var S,N,K,Z: String;

      A,B,C: LongInt;
      P,cod: Integer;
Begin
  Write('Введите выражение: ');

  Readln(S);

  P:=Pos('+',S);

  If P<>0 Then Z:='+'

     Else Begin

            P:=Pos('-',S);

            Z:='-';

          End;

  N:=Copy(S,1,P-1);
  Val(N,A,cod);
  K:=Copy(S,p+1,Length(S)-P));

  Val(K,B,cod);

  If Z='+' Then C:=A+B

           Else C:=A-B;

  Writeln(A,Z,B,'=',C);

End.

Упражнения:  
Напишите программы решения следующих задач:

1. Задан текст. Проверить, является ли он десятичной записью:

1) целого числа; 



2) вещественного числа.
2. Заданный текст представляет собой последовательность цифр 0 и 1: a1a2a3…an. Зашифровать текст, получив новый текст b1b2b3…bn , где b1=a1, 

3. Удалить из заданного текста все цифры, сохранив порядок записи.

4. Вставить в заданном тексте после каждого знака, который не является цифрой, его порядковый номер; например, для текста 'Школа-гимназия 23' нужно получить:

'Ш1к2о3л4а5-6г7и8м9н10а11з12и13я14 1523'.

